Acessando Sites da Web através de VBA – Pt.01
Esta é a primeira parte de uma série de artigos sobre acessar Página da Internet através de código VBA. Vamos lá...
[bookmark: more]Primeiro você deve se certificar de que tenha a referência “Microsoft Internet Control” em seu VBE (Visual Basic Editor).
[image: http://4.bp.blogspot.com/-A11UcRpAGKs/TVgd0XHvoII/AAAAAAAAADY/vgsvRko5HcY/s1600/Imagem1.png]
Para acessar o VBE, no Excel 2007 ou 2010, tecle a combinação ALT + F11 ou clique em Visual Basic na guia Desenvolvedor.
[image: http://3.bp.blogspot.com/--N7gEFaQRwA/TVgd5xlr1sI/AAAAAAAAADc/i4xbunPDKgc/s1600/Imagem2.png]
Para acessar o VBE, no Excel 2003 e inferiores, tecle a combinação ALT + F11 ou clique em Ferramentas -> Macro -> Editor do Visual Basic.
[image: http://2.bp.blogspot.com/-rYfLJ6AwOSA/TVgeBzLCWYI/AAAAAAAAADg/8J02qf-e0_Y/s1600/Imagem3.png]
Para acessar a caixa de referências do projeto, no VBE, clique em Ferramentas -> Referências.
[image: http://4.bp.blogspot.com/-7H4E0PiLf6k/TVgeDfgqEoI/AAAAAAAAADk/c7nh_R0QxzM/s1600/Imagem4.png]
Bem, vamos aos primeiros passos.

Crie um módulo novo (No VBE, Inserir -> Módulo) e digite o seguinte código:


Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer  
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
End Sub

Iniciamos com uma rotina simples, na qual se declara uma variável do tipo Internet Explorer, em seguida instancia-se ela (New).

Você vai reparar que quando se cria uma instância, o navegador não é aberto, mas o processo dele aparece nos Processos do Windows (CTRL + SHIFT + ESC).

Usamos o comando Navigate, que recebe como parâmetro a URL do site que você deseja abrir. Esse comando é o equivalente ao botão Ir do Internet Explorer.

E para finalizar, usamos o comando Visible = True para que o navegador seja exibido na tela.

Bom, por enquanto é só. Aguardem a segunda parte onde explicarei mais coisas que são possíveis de serem feitas através da classe Internet Explorer. Para vocês terem uma idéia, é possível mudar o método GET para POST de um site (de forma local e não no servidor do próprio site), enviar instruções Java Script, ler dados de campos de formulários e tabelas, navegar entre os frames e etc. 

Ah, e para aqueles que por algum motivo não conseguiram declarar a variável como Internet Explorer, segue caminho alternativo.


Sub AcessaPagina_()

    Dim ie As Object

    Set ie = CreateObject("InternetExplorer.Application")

    ie.Navigate "http://www.google.com.br"

    ie.Visible = True

End Sub

Acessando Sites da Web através de VBA – Pt.02
Bem pessoal, nessa segunda parte de meu artigo sobre acesso à uma página da Web pelo VBA do Excel, quero ensinar como clicar em botões e preencher caixas de texto com valores.

Apenas revisando o código do nosso artigo anterior:
 
Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer  

    ie.Navigate "http://www.google.com.br"

    ie.Visible = True

End Sub

Nós vimos que este código nada mais faz do que abrir a página do Google.

Preencher dados de uma caixa de texto

O truque é você descobrir qual é o nome do campo (propriedade name da tag HTML). Claro, para isso você tem que ter um conhecimento intermediário de HTML e um pouquinho de conhecimento sobre JavaScript ajudaria também.

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate http://www.google.com.br

    ie.Document.All.Item("q").innertext = "Pesquisa"
    ie.Visible = True

End Sub

“q” é o nome da caixa de texto do site do Google onde você digita a sua pesquisa.

Document é nome do objeto que representa o documento da página.

All é a propriedade onde você acessa todos os objetos do documento, tais como links, tabelas, imagens, botões, formulários e por ai vai.

Item é uma propriedade onde eu defino qual objeto do documento (links, forms e etc.) eu desejo selecionar, você pode especificar tanto o nome do objeto como o número que ele representa no site.

innerText é uma propriedade específica do objeto que você está acessando onde você manipula o texto dentro do objeto.

Uma boa dica para descobrir o nome destes objetos e tudo o mais é utilizar o Local Variables ou Variáveis Locais do Visual Basic Editor. No VBE, clique em Exibir -> Janela Variáveis Locais.
[image: http://2.bp.blogspot.com/-P1q4eI3UV50/TWAbRMkS_jI/AAAAAAAAAEQ/DkOjK9PKErg/s1600/Imagem1.png]
Acessando janela de Variáveis Locais
[image: http://3.bp.blogspot.com/-OMUx9njst90/TWAbSRVZoWI/AAAAAAAAAEU/1aS5Hy3dgL0/s1600/Imagem2.png]
Janela de Variáveis Locais

Outra maneira de preencher uma caixa de texto é:


Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://www.google.com.br"

    ie.Document.getElementsByName("q").Item.InnerText = "Pesquisa"

    ie.Visible = True

End Sub

getElementsByName é um método onde você retorna o objeto pelo nome que ele possui.

Questão: E se o objeto que eu quero atribuir valor possui a propriedade id e eu quero referenciar por esta?
Simples, use o método getElementById:

ie.Document.getElementById("id").Item.Value = "Valor"


Clicando no botão

 
Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://www.google.com.br"

    ie.Document.getElementsByName("q").Item.InnerText = "Pesquisa"

    ie.Document.all.Item("btnG").Click

    ie.Visible = True

End Sub

Use o método Click.
Claro, funciona da outra maneira que expliquei também:


Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://www.google.com.br"

    ie.Document.getElementsByName("q").Item.InnerText = "Pesquisa"

    ie.Document.getElementsByName("btnG").Item.Click

    ie.Visible = True
	
End Sub
Obs.: pode usar o método getElementById que funciona também.

Ah, e se você tentou executar de uma só vez todo o código e obteve um erro assim:
[image: http://1.bp.blogspot.com/-WZN7ryn6EIM/TWAchnmKKDI/AAAAAAAAAEY/2EF6_0Q0NAA/s1600/Imagem3.png]
É porque o código tentou preencher a caixa de texto “Pesquisa” do site do Google antes do site ser carregado.

Como resolver?
Inclua um loop enquanto o site estiver carregando.


Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://www.google.com.br"

    While ie.Busy

    Wend
  
    ie.Document.getElementsByName("q").Item.InnerText = "Pesquisa"

    ie.Document.getElementsByName("btnG").Item.Click

    ie.Visible = True
 
End Sub
Busy é uma propriedade que fica true enquanto a página está carregando e false quando ela já estiver totalmente carregada.

Acessando Sites da Web através de VBA – Pt.03
Continuando o artigo sobre acesso a sites pelo VBA, quero falar, hoje, sobre como selecionar valores em um CheckBox, Radio Button, ComboBox e ListBox nas páginas da Web através de VBA.
Vamos usar como exemplo um site próprio que criei hospedado no NoFeeHost.
Clique aqui para acessar o site.

[image: https://lh3.googleusercontent.com/-zDdnjoQIGlc/TWpqGxHZfMI/AAAAAAAAAEs/moRyEn8NKas/s1600/1298819583_Checkbox+Full.png]Selecionando valores no CheckBox


Existem diversas formas, mas vamos à mais simples por enquanto:

Sub AcessaPagina()

 Dim ie As InternetExplorer

 Set ie = New InternetExplorer

 ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

 ie.Visible = True

 ie.Document.All.Item("vehicle").Item(1).Checked = True

End Sub

Este código seleciona o segundo item. Vamos a explicação:

vehicle é o nome do grupo do CheckBox, ou melhor, da propriedade name do CheckBox.

Item(1) é o item que eu desejo selecionar. O número 1 (um) significa o segundo item, pois a neste caso o primeiro item seria o 0 (zero).

Se você olhar o código-fonte da página, verá o seguinte:

<input type="checkbox" name="vehicle" value="Bike" /> I have a bike<br />
<input type="checkbox" name="vehicle" value="Car" /> I have a car 

Outra forma é omitindo a propriedade Item().


Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True
 
    ie.Document.All.Item("vehicle")(1).Checked = True
End Sub

E claro o que ensinei na segunda parte deste artigo se aplica aqui também:


Sub AcessaPagina()
    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    ie.Document.getElementsByName("vehicle")(1).Checked = True
 
End Sub


[image: https://lh4.googleusercontent.com/-ysjZADDZJdY/TWpqhoDHWCI/AAAAAAAAAEw/unhrWQ__H3s/s1600/1298819699_radiobutton_yes.png]Selecionando valores no Radio Button

É o mesmo código utilizado quando se seleciona no CheckBox.

Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate http://users5.nofeehost.com/BackOfficeSite/

    ie.Visible = True

    ie.Document.All.Item("sex").Item(1).Checked = True

End Sub


[image: https://lh5.googleusercontent.com/-08fpTqu0FGA/TWprCtVUspI/AAAAAAAAAE0/kiidbhit-Cw/s1600/1298819837_stock_form-combobox.png]Selecionando valores no ComboBox e ListBox 

O código fonte do ComboBox (ou dropdown list), na página web, é o seguinte:

<select name="cars">
        <option value="volvo">Volvo</option>
        <option value="saab">Saab</option>
        <option value="fiat">Fiat</option> 
        <option value="audi">Audi</option>
</select>

Agora, vamos ao código VBA:

Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    ie.Document.All.Item("cars").Item(1).Selected = True

End Sub

É quase o mesmo código utilizado para CheckBox e RadioButton. A diferença é que utilizo a propriedade Selected ao invés de Checked.

Para o ListBox não existe segredo, pois ele usa o mesmo código fonte HTML para o Combobox. A diferença é a propriedade size do código fonte HTML que é alterada. Veja o código fonte HTML do ListBox:

<select name="drop1" id="Select1" size="4">
        <option value="1">item 1</option>
        <option value="2">item 2</option>
        <option value="3">item 3</option>
        <option value="4">item 4</option>
        <option value="0">All</option>
</select>

O código VBA ficaria da seguinte forma:

Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    ie.Document.All.Item("drop1").Item(1).Selected = True

End Sub  


Outra questão: e quando meu ListBox permite múltipla seleção de valores? Simples: acrescente mais uma linha de código selecionando o valor:
Sub AcessaPagina()


    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    ie.Document.All.Item("drop1").Item(0).Selected = False

    ie.Document.All.Item("drop1").Item(1).Selected = True

    ie.Document.All.Item("drop1").Item(3).Selected = True

End Sub  
[image: https://lh3.googleusercontent.com/-VGFBPuBH3cI/TWp_AgUmCSI/AAAAAAAAAFE/CHsvVUROfxM/s1600/ListBox.png]
O código fonte HTML de um ListBox com múltipla seleção é o seguinte:

      <select name="drop1" id="Select1" size="4" multiple="multiple">
        <option value="1">item 1</option>
        <option value="2">item 2</option>
        <option value="3">item 3</option>
        <option value="4">item 4</option>
        <option value="0">All</option>
      </select>


[image: https://lh3.googleusercontent.com/-suSqtDPItwo/TWp66om6mNI/AAAAAAAAAFA/Byw_YipdOV4/s1600/Task-List-icon.png] Melhores práticas para selecionar valores no CheckBox, RadioButton, ComboBox e ListBox


Você há de convir comigo que os códigos usados acima não são nenhum um pouco seguro. Quero dizer que, referenciar o item do CheckBox, RadioButton, ListBox ou ComboBox pelo número que ele representa (item(1), por exemplo) é muito arriscado. Pode acontecer de o desenvolvedor do site alterar a ordem dos itens no ComboBox ou o item que você está selecionando não existir mais.

Para nos precavermos disso, utilizaremos o seguinte código: 

Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    Dim obj As Object

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    For Each obj In ie.Document.All.Item("drop1").Options
        If obj.innertext = "item 3" Then
            obj.Selected = True
        End If
    Next obj

End Sub  

Estou percorrendo todas as opções do ListBox e verifico se o texto corresponde ao que eu quero selecionar. Caso positivo eu o seleciono. Este código se aplica ao ComboBox também.

Agora, para o RadioButton e CheckBox usamos o seguinte código:

Sub AcessaPagina()

    Dim ie As InternetExplorer

    Set ie = New InternetExplorer

    Dim obj As Object

    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/"

    ie.Visible = True

    For Each obj In ie.Document.All.Item("sex")
        If obj.Value = "female" Then
            obj.Checked = True
        End If
    Next obj

End Sub

O que muda é que não uso a propriedade innerText para comparar valores, mas a propriedade Value e não utilizo nenhuma propriedade indicando subitens do All.Item("sex"), mas referencio direto. Quero dizer com isso, não utilizo algo do tipo All.Item("sex").Options ou All.Item("sex").Items (não existem e gera erro se você tentar desta forma).
É isso ai. Por enquanto é só. Aguardem a quarta parte deste artigo. Ainda tem muita coisa a ser explorada como por exemplo clicar em links, como lidar com frames, comandos javascripts, janelas de interação do navegador (inputbox, confirm e etc.) e por ai vai.

Acessando Sites da Web através de VBA – Pt.04
Nesta quarta parte do meu artigo sobre acesso a sites da Web através de VBA, vou abordar os seguintes assuntos: clicando em links, lendos dados de tabelas e lidando com frames.

[image: https://lh4.googleusercontent.com/-jUosdXdZGgU/TXLyDBLRCOI/AAAAAAAAAFM/WQv9Mk5XIlk/s1600/1299378681_insert-hyperlink.png]Clicando em Links


Utilizaremos, como exemplo, uma página simples que criei no host NoFeeHost.
Clique aqui para acessar.

O código fonte html de um hyperlink é o seguinte:

<a href="http://www.yahoo.com/">Click Here</a> 


O código VBA se resume em utilizar o objeto Links que está contido no objeto Document:

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Links.htm"
    ie.Visible = True
    ie.Document.Links(0).Click

End Sub

Uma dica que posso dar é utilizar a propriedade Length para verificar se eu tenho links no meu site. Assim eu evito erros em tempo de execução.
Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Links.htm"
    ie.Visible = True
    
    If ie.Document.Links.Length > 0 Then
        ie.Document.Links(0).Click
    End If

End Sub

E se a página que estou acessando tiver milhares de links? Bem, então utilize uma estrutura de repetição que saia iterando entre cada link.

Sub AcessaPagina()


    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Links.htm"
    ie.Visible = True
    
    Dim obj As Object
    
    For Each obj In ie.Document.Links
        If obj.href = "http://www.yahoo.com/" Then
            obj.Click
        End If
    Next obj

End Sub

Na segunda parte deste artigo eu mencionei sobre o uso da Janela de Variáveis Locais (Local Variables Window). Torno a repetir chamando a atenção para a variável "obj" do exemplo acima. Veja quantas propriedades e quantas informações podem extrair:
[image: https://lh4.googleusercontent.com/-nrK0JkMe7nQ/TXLwtS4cMJI/AAAAAAAAAFI/R-vDaF4cxKg/s1600/local_variable_window.PNG]
Janela de Variáveis Locais
É possível vermos a porta que é utilizada para acesso (80), o protocolo (http), o texto contido dentro das tags do link ("Click Here") , entre outros. Quero dizer que, você pode se utilizar destas informações para outros propósitos bastando utilizar o nome que aparece na janela de variáveis locais, exemplo:

Dim varProtocolo as String
varProtocolo = ie.document.links(0).protocol
[image: https://lh5.googleusercontent.com/-FDra4U1TiNY/TXLysrxgKmI/AAAAAAAAAFQ/jgET24oL1ic/s1600/1299378852_table.png]Lendo dados de tabelas

Clique aqui para acessar o site que utilizaremos como exemplo.

O código fonte html de uma tabela é o seguinte:

<table>
     <tr>
          <td>Conteúdo de célula 1 da linha 1</td> 
          <td>Conteúdo de célula 2 da linha 1</td> 
          <td>Conteúdo de célula 3 da linha 1</td>  
     </tr> 
     <tr>
          <td>Conteúdo de célula 1 da linha 2</td> 
          <td>Conteúdo de célula 2 da linha 2</td> 
          <td>Conteúdo de célula 3 da linha 2</td>  
     </tr>  
</table> 

Acessar os dados de tabelas não é tão simples como tenho apresentado até agora. É importante você entender o modelo de objeto de uma tabela.
[image: https://lh4.googleusercontent.com/-CCPe224hrtI/TXL5qoKvINI/AAAAAAAAAFk/KNV9-cH91dk/s1600/desenho_tabela.png]
Imagem para entendimento das propriedades de uma tabela
Agora, vamos a um primeiro exemplo de código:
Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Tables.htm"
    ie.Visible = True
    
    MsgBox ie.Document.all.Item(6).Rows(0).Cells(0).innerText

End Sub 

Item(6) = na coleção de itens em meu site, o item 6 (que é o sétimo item dado que a contagem começa a partir de 0) corresponde à uma tabela (em html, <table></table>);
Rows(0) = primeira linha de minha tabela (em html <tr></tr>);
Cells(0) = primeira célula de minha tabela (em html <td></td>);
innerText = retorna o texto dentro da célula.

Concorda que esse código é bem rudimentar (e perigoso). E se meu site tiver mais de 1000 itens, como fazer para saber qual é a tabela que quero acessar? Terei que sair caçando na Janela de Variáveis Locais qual item corresponde à tabela? Não! Graças ao bom Deus, aprendi uma maneira de referenciar a tabela sem precisar saber qual o número do item a qual ela corresponde.

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Tables.htm"
    ie.Visible = True
        
    Dim elemCollection As Object
    Dim obj As Object
    
    Set elemCollection = ie.Document.getElementsByTagName("TABLE")
    
    For Each obj In elemCollection
    
        MsgBox obj.Rows(0).Cells(0).innerText

    Next obj

End Sub

Utilizamos o método getElementsByTagName(nome_da_tag_html) para retornar a coleção de objetos correspondentes à tag <table>, ou seja, se tivéssemos mais de uma tabela em nossa página, esse método retornaria todas atribuindo-as à variável elemCollection. Então, através da variável obj, iteraríamos (loop) por todas as tabelas retornadas na variável elemColletion. E com a variável obj, acessamos as linhas (rows) e células (cells) da tabela.
Esse método também serve para todas outras tags, tais como "a", "tr", "td" e etc.
E para finalizar o assunto de tabelas, segue algoritmo que lê os dados da tabela e os lança na planilha:
Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Tables.htm"
    ie.Visible = True
        
    Dim elemCollection As Object
    
    Dim t As Integer
    Dim r As Integer, c As Integer
    
    Set elemCollection = ie.Document.getElementsByTagName("TABLE")
    
    For t = 0 To elemCollection.Length - 1
        For r = 0 To elemCollection(t).Rows.Length - 1
            For c = 0 To elemCollection(t).Rows(r).Cells.Length - 1
                ThisWorkbook.Worksheets(1).Cells(r + 1, c + 1) = elemCollection(t).Rows(r).Cells(c).innerText
            Next c
        Next r
    Next t

End Sub

[image: https://lh5.googleusercontent.com/-yiu_1ReNgVA/TXME7zHKHsI/AAAAAAAAAFs/2k3ybrdZ75g/s1600/1299383504_stock_insert-floating-frame.png]Lidando com Frames
Bem, lidar com Frames é extremamente simples. Basta você saber o nome (propriedade "name" na tag html) do frame onde você quer acessar o conteúdo.
Clique aqui para acessar o site que estou utilizando como exemplo.
O código fonte de um frame é o seguinte:
<frameset cols="25%,75%">
     <frame src="http://users5.nofeehost.com/BackOfficeSite/Frame_A.html" name="Frame_A"></frame>
     <frame src="http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Tables.htm" name="Frame_B"></frame>
</frameset>
Vamos ao código VBA para acessar:
Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    ie.Navigate "http://users5.nofeehost.com/BackOfficeSite/Vba_Site_Iv_Frames.htm"
    ie.Visible = True
        
    MsgBox ie.Document.frames("Frame_A").Document.all.Item(0).innertext

End Sub

frames(nome_do_frame ou numero_do_item) = acessa determinado frame de acordo com o nome especificado ou com o número correspondente (frames(0), por exemplo).
Acessando Sites da Web através de VBA – Pt.05
Olá Galera, hoje finalizarei este artigo sobre acesso à sites da Web pelo VBA.
Hoje vou falar sobre execução de scripts JavaScript, rotina de tratamento para carregamento de página e dar algumas dicas.

[image: https://lh6.googleusercontent.com/-Eor6LbDXJ60/TXvzn-hHTOI/AAAAAAAAAF8/a87jK8ibR0c/s1600/1299968880_application-javascript.png]Executando Scripts JavaScript

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
   
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
   
    ie.Document.parentwindow.execscript ("alert('Hellow World')")

End Sub

O exemplo acima é bem simples.

Você pode se perguntar: para que executarei instruções javascript? Por vários motivos.

Imagine que você tem um site onde existe validação de dados. Você pode chamar o método de gravação diretamente ao invés de mandar clicar no botão "Gravar" e assim "furar" a validação de dados, por exemplo.

    ie.Document.parentwindow.execscript ("metodo_gravar()")

Você pode habilitar/desabilitar campos.
Vamos brincar um pouco com a página do Google:

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    'Disabilita a caixa de texto de pesquisa
    ie.Document.parentwindow.execscript ("document.forms[0].q.disabled=true")
    
    'Muda o texto do botão de "Pesquisa Google"
    ie.Document.parentwindow.execscript ("document.forms[0].btnG.value='É nois'")

End Sub

Enfim, uma variedade de coisas. Agora, você tem uma lição de casa: aprender javascript.

Apenas uma observação importante: toda essa brincadeira de mudar as características da página, roda apenas localmente (cliente) e não no servidor. Se você atualizar a página (F5) verá que as mudanças desaparecerão.

Rotina de tratamento para carregamento de página

Abaixo alguns códigos para tratar o carregamento de página.
Sub AcessaPagina()
    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    Do While ie.Busy
    Loop
End Sub

Outra rotina:
Sub AcessaPagina()
    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    Do Until ie.Document.ReadyState = "complete"
    Loop
End Sub


Outra rotina:
Sub AcessaPagina()
    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    Do Until ie.ReadyState = READYSTATE_COMPLETE
    Loop
End Sub


O que diferencia? Em alguns momentos o objeto InternetExplorer sinalizará que o carregamento está completo, mas pode acontecer de uma parte do documento ou o frame não estiver totalmente carregado. Então você pode utilizar uma das 3 rotinas acima sugeridas.

[image: https://lh4.googleusercontent.com/-YK5sT3xU_Rg/TXv4XH1EaAI/AAAAAAAAAGE/s98-zxOjV7U/s1600/1299970096_ui-status-bar-blue.png]Mudando a Barra de Status


Sub AcessaPagina()
    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    ie.StatusBar = "Mudei a barra de status!"
End Sub

[image: https://lh3.googleusercontent.com/--Ec5fQ7gWJA/TXv5kPj7X8I/AAAAAAAAAGI/Sf4Q9oThrHY/s1600/1299970435_dialog-close.png]Fechando o navegador


Sub AcessaPagina()
    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://www.google.com.br"
    ie.Visible = True
    
    ie.Quit
    Set ie = Nothing
End Sub

Bom pessoal, por enquanto é isso.
Num futuro não muito distante falarei mais sobre acessar páginas da Web através de VBA.

Usando o método ExecWB
Olá amigos desenvolvedores VBA.
Vamos falar, hoje, novamente de Internet + VBA.
Este método executa instruções no navegador, tais como Imprimir, Salvar como, mudar Zoom e etc.

Sintaxe:
ExecWB(cmdID As OLECMDID, cmdexecopt As OLECMDEXECOPT, [pvaIn As Variant], [pvaOut As Variant])

cmdID As OLECMDID = Valor Enumerado que define qual comando você quer executar. Abaixo segue tabela dos comandos.

  OLECMDID_OPEN                            = 1,
  OLECMDID_NEW                             = 2,
  OLECMDID_SAVE                            = 3,
  OLECMDID_SAVEAS                          = 4,
  OLECMDID_SAVECOPYAS                      = 5,
  OLECMDID_PRINT                           = 6,
  OLECMDID_PRINTPREVIEW                    = 7,
  OLECMDID_PAGESETUP                       = 8,
  OLECMDID_SPELL                           = 9,
  OLECMDID_PROPERTIES                      = 10,
  OLECMDID_CUT                             = 11,
  OLECMDID_COPY                            = 12,
  OLECMDID_PASTE                           = 13,
  OLECMDID_PASTESPECIAL                    = 14,
  OLECMDID_UNDO                            = 15,
  OLECMDID_REDO                            = 16,
  OLECMDID_SELECTALL                       = 17,
  OLECMDID_CLEARSELECTION                  = 18,
  OLECMDID_ZOOM                            = 19,
  OLECMDID_GETZOOMRANGE                    = 20,
  OLECMDID_UPDATECOMMANDS                  = 21,
  OLECMDID_REFRESH                         = 22,
  OLECMDID_STOP                            = 23,
  OLECMDID_HIDETOOLBARS                    = 24,
  OLECMDID_SETPROGRESSMAX                  = 25,
  OLECMDID_SETPROGRESSPOS                  = 26,
  OLECMDID_SETPROGRESSTEXT                 = 27,
  OLECMDID_SETTITLE                        = 28,
  OLECMDID_SETDOWNLOADSTATE                = 29,
  OLECMDID_STOPDOWNLOAD                    = 30,
  OLECMDID_FIND                            = 32,
  OLECMDID_DELETE                          = 33,
  OLECMDID_HTTPEQUIV_DONE                  = 35,
  OLECMDID_ENABLE_INTERACTION              = 36,
  OLECMDID_ONUNLOAD                        = 37,
  OLECMDID_PROPERTYBAG2                    = 38,
  OLECMDID_PREREFRESH                      = 39,
  OLECMDID_SHOWSCRIPTERROR                 = 40,
  OLECMDID_SHOWMESSAGE                     = 41,
  OLECMDID_SHOWFIND                        = 42,
  OLECMDID_SHOWPAGESETUP                   = 43,
  OLECMDID_SHOWPRINT                       = 44,
  OLECMDID_CLOSE                           = 45,
  OLECMDID_ALLOWUILESSSAVEAS               = 46,
  OLECMDID_DONTDOWNLOADCSS                 = 47,
  OLECMDID_UPDATEPAGESTATUS                = 48,
  OLECMDID_PRINT2                          = 49,
  OLECMDID_PRINTPREVIEW2                   = 50,
  OLECMDID_SETPRINTTEMPLATE                = 51,
  OLECMDID_GETPRINTTEMPLATE                = 52,
  OLECMDID_PAGEACTIONBLOCKED               = 55,
  OLECMDID_PAGEACTIONUIQUERY               = 56,
  OLECMDID_FOCUSVIEWCONTROLS               = 57,
  OLECMDID_FOCUSVIEWCONTROLSQUERY          = 58,
  OLECMDID_SHOWPAGEACTIONMENU              = 59,
  OLECMDID_ADDTRAVELENTRY                  = 60,
  OLECMDID_UPDATETRAVELENTRY               = 61,
  OLECMDID_UPDATEBACKFORWARDSTATE          = 62,
  OLECMDID_OPTICAL_ZOOM                    = 63,
  OLECMDID_OPTICAL_GETZOOMRANGE            = 64,
  OLECMDID_WINDOWSTATECHANGED              = 65,
  OLECMDID_ACTIVEXINSTALLSCOPE             = 66,
  OLECMDID_UPDATETRAVELENTRY_DATARECOVERY  = 67

cmdexecopt As OLECMDEXECOPT = Valor Enumerado que define como você quer que esse comando seja executado, exibindo uma janela de prompt para o usuário e etc. Mais detalhes abaixo.

  OLECMDEXECOPT_DODEFAULT       = 0,
  OLECMDEXECOPT_PROMPTUSER      = 1,
  OLECMDEXECOPT_DONTPROMPTUSER  = 2,
  OLECMDEXECOPT_SHOWHELP        = 3

[pvaIn As Variant] = Parâmetros de entrada. Pode ser numérico ou alfanumérico. Dependendo do comando que você estiver executando esse campo poderá ser preenchido (SAVEAS, por exemplo) ou poderá ficar em branco (VbNullString). 

[pvaOut As Variant] = Parâmetros de saída.

Vamos a alguns exemplos:

Mudar o tamanho das letras da página
Sub AcessaPagina()

Dim ie As InternetExplorer
Set ie = New InternetExplorer

ie.Navigate "http://www.google.com.br/#hl=pt-BR&source=hp&biw=1276&bih=837&q=uol+not%C3%ADcias&aq=f&aqi=g10&aql=&oq=&fp=516cc978868108d7"

ie.Visible = True 

ie.ExecWB OLECMDID_ZOOM, OLECMDEXECOPT_DONTPROMPTUSER, CLng(1), vbNull

End Sub

Obs.: Veja que no parâmetro de entrada especificamos o valor 1, que é a menor letra possível. Você pode especificar de 1 até 4.

Exibir janela de impressão
[...]
ie.ExecWB OLECMDID_PRINT, OLECMDEXECOPT_DODEFAULT
[...]

Exibir janela de configurar página
[...]
ie.ExecWB OLECMDID_PAGESETUP, OLECMDEXECOPT_DODEFAULT
[...]

Exibir janela de salvar como
[...]
ie.ExecWB OLECMDID_SAVEAS, OLECMDEXECOPT_DONTPROMPTUSER, "Teste2.html", vbNullString
[...]

Imprimir
[...]
ie.ExecWB OLECMDID_PRINT, OLECMDEXECOPT_DONTPROMPTUSER, "", ""
[...]
Mudando o método POST de um formulário para GET

Voltando a falar sobre Internet + VBA, hoje eu vou ensinar um truque interessante.

Uma breve explicação sobre GET e POST

POST envia (submit) as informações dos campos do formulário de sua página HTML através de uma transação HTTP.

GET anexa as informações dos campos do formulário de sua página HTML na URL (na barra de endereços do seu navegador aparecerão os nomes dos campos e os valores preenchidos).

Ver: http://www.w3schools.com/jsref/prop_form_method.asp.

Trocando o método de envio do formulário para GET ou POST 

Lembrando que as alterações são locais (client) e não no servidor (server), portanto se você atualizar a página, as modificações serão desfeitas.

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
   
    ie.Navigate "https://www.evernote.com/Login.action?targetUrl=%2FHome.action"
    ie.Visible = True
   
    Do While ie.Busy
    Loop
   
    ie.Document.all.Item("username").Value = "teste"
    ie.Document.all.Item("password").Value = "teste"
   
    ie.Document.Forms(0).Method = "GET"
    ie.Document.Forms(0).Submit
   
End Sub

Vejam que as informações apareceram na barra de endereços do navegador:

[image: http://4.bp.blogspot.com/-yALbx3OCc9A/TdAe8qo42wI/AAAAAAAAAJw/A4iBxX17frY/s1600/get001.GIF]
Usando método POST

Agora, se mudarmos para POST, como fica?

[...]
    ie.Document.Forms(0).Method = "GET"
    ie.Document.Forms(0).Submit
[...]

[image: http://4.bp.blogspot.com/-l_oqvMAAE6o/TdAffkWo2ZI/AAAAAAAAAJ0/ArDSw3g0AGI/s1600/get002.GIF]
Usando método GET

É isso ai. Em breve ensinarei como fazer download de arquivo via código VBA e para isso você precisará de uma URL para dizer de onde vem o download. Muitas vezes, quando você clica em "download" em determinado site, por exemplo, a URL do download fica mascarada (POST), mas com este método você consegue enxergar qual é a URL e fazer download diretamente (ou via código VBA ou colando a URL na barra de endereços do navegador).

Exemplo de utilização do método Execscript

Olá pessoal, em um dos meus artigos eu ensinei a usar instruções javascript apartir de seu código VBA.

Hoje, quero dar um exemplo prático do poder desta ferramenta.

Imagine que eu quero fazer um download no Freakshare.com

[image: http://3.bp.blogspot.com/-1tby2hek0Lc/TdAijWmRpeI/AAAAAAAAAJ4/DinP5WnJoGQ/s1600/freakshare001.GIF]
Site Freakshare.com

Veja na imagem, que existe um contador de tempo que geralmente se inicia em 59 segundos.

O código para abrir essa página seria assim:

Sub AcessaPagina()


    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://freakshare.com/files/cnv4vonx/-Claymore-Vol_14.rar.html"
    ie.Visible = True


    Do While ie.Busy
    Loop


End Sub

Se olharmos com mais detalhe o código fonte do site Freakshare.com, observaremos o seguinte:
[image: http://4.bp.blogspot.com/-kwn0G3JLbmI/TdAjNCbxdkI/AAAAAAAAAJ8/fNwCdhkDdfA/s1600/freakshare002.GIF]
Código do site Freakshare.com

Com execução javascript, você pode alterar esse valor da variável time, mas já adianto que não retardará o download, muito provável porque do lado servidor do site Freakshare existe uma validação.

O que vale é a brincadeira e o conhecimento aprendido, certo?
  
Como ficaria o código, então?

Sub AcessaPagina()

    Dim ie As InternetExplorer
    Set ie = New InternetExplorer
    
    ie.Navigate "http://freakshare.com/files/cnv4vonx/-Claymore-Vol_14.rar.html"
    ie.Visible = True

    Do While ie.Busy
    Loop
    
    Do Until ie.Document.ReadyState = "complete"
    Loop
    
    ie.Document.parentwindow.execscript ("time = 5")

End Sub

Execute esse código e veja o que acontece!

Se você tentar fazer o download após executar este código e digitar o Captcha no site, receberá a seguinte mensagem do site Freakshare.com

[image: http://3.bp.blogspot.com/-0y1Ogom7boA/TdAmMbq9yMI/AAAAAAAAAKA/4jf7kLd1FHA/s1600/freakshare003.GIF]
Site Freakshare.com

Essa é a validação da qual eu falei anteriormente.

Bom, espero que você tenha aprendido algo e use com sabedoria esse truque.

Lidando com caixas de mensagens

Uma das encruzilhadas que muitos programadores encontram ao trabalhar com VBA e Internet é como lidar com caixas de mensagens no navegador. 

[image: http://4.bp.blogspot.com/-NBXSI6FJPck/TdlWSyL1hVI/AAAAAAAAAKE/alwSVNqogiA/s1600/001.GIF]
Caixa de mensagem malvada

Cedo descobrem que o objeto do Internet Explorer não oferece nada (nem uma pista) para poder lidar com essas caixas de mensagens. Então, num frenesi, os programadores tentam de tudo para fechar essa maldita caixa de mensagem: SendKeys, comandos que nem o próprio dono do código sabe explicar, macumba, mãe Diná, sabonete do descarrego e por ai vai... E nada funciona!

Qual é a solução então? Você terá de recorrer às APIs do Windows.
Para o nosso caso descrito acima teremos de utilizar duas APIs: SendMessage e FindWindow.

Não vou me prender a explicar detalhes de cada APIs (nem o que venha a ser uma API do Windows), mas basicamente FindWindow localiza a janela que você quer fechar (ou outros comandos) e SendMessage envia o comando (no caso, "fechar") para a janela que você encontrou através do FindWindow.

No começo do código declare as seguintes linhas:

Public Declare Function SendMessage Lib "User32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long
 
Public Declare Function FindWindow Lib "User32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long
 
Public Const WM_CLOSE = &H10

OK, agora você pode começar a usar as APIs.

A constante WM_CLOSE representa o comando "fechar" (o CLOSE é bem sugestivo, certo?). Você utilizará essa constante como parâmetro na chamada da API SendMessage.

E agora o código completo:

Sub AcessaPagina()

Dim winHwnd As Long

'[...] Bloco de código que acessa a página, até o momento em que abre a mensagem

winHwnd = FindWindow(vbNullString, "Mensagem da página da Web") 

If winHwnd <> 0 Then
     SendMessage winHwnd, WM_CLOSE, 0&, 0&
End If

End Sub

Explicando brevemente, winHwnd é uma variável do tipo Long que receberá um número inteiro que representará a janela que você está retornando pela API FindWindow.

Importante notar que a API FindWindow busca a janela pelo texto que aparece na barra de títulos dela. Veja a imagem no começo deste artigo e notará que o texto da barra de títulos é Mensagem da página da Web. Essa mensagem varia de acordo com o site ou versão de navegador.

E por fim utilizamos a API SendMessage dizendo para ela fechar a janela encontrada.

